

PrácticaJP3

Práctica 3: Funciones y gráficas. Límites y continuidad. Derivación.

3.1. Definición de funciones

En la práctica explicamos brevemente cómo definir nuestras propias funciones. Recordemos:

```
f(x)=x^2-3*x+1
a=var('a')
f(1); f(a+1); expand(f(a+1))

-1
(a + 1)^2 - 3*a - 2
a^2 - a - 1
```

Si una función tiene más de un argumento, éstos se separan por comas.

```
y=var('y'); g(x,y)=x+y*x^2-2
g(3,6)

55
```

3.2. Dibujo de gráficas de funciones

Para la representación de funciones reales de variable real se usa la orden


`plot(f(x), xmin, xmax)` dibuja la función **f** con **x** entre **xmin** y **xmax**.

En una misma gráfica podemos representar más de una función.

Veamos algunos ejemplos:

1) Dibujamos ahora la función $y=\cos x$


```
plot(cos(x), -2*pi, 2*pi)
```



En la gráfica podemos intuir que su dominio es todo \mathbf{R} y que es periódica.


2) Dibujamos a la vez las funciones $y=\sin x$, $y=\cos x$, en distinto color.

```
plot(sin(x), -2*pi, 2*pi, rgbcolor=(1,0,0))+plot(cos(x), -2*pi, 2*pi, rgbcolor=(0,1,0))
```



3) También podemos definir funciones definidas por nosotros mismos. Por ejemplo:

```
g(x)=arctan(1/(x-1)); plot(g(x), -2, 2)
```



La gráfica tiene un defecto, ha unido con un segmento los puntos de discontinuidad, así que se puede llevarnos al error de pensar que es una función continua. Esto sucede porque el método de dibujar de SAGE es muestrear unos cuantos puntos y luego unir con segmentos, así que todo lo pinta continuo.


Aparte de escribir correctamente la sintaxis de la orden `plot`, tenemos que **elegir el intervalo** en el que dibujamos. Los **errores más frecuentes** al respecto son:

Elegir un intervalo donde no está definida la función; es decir, que no está en el dominio:

4) Dibujando $\log(x)$ con x en $[-2,2]$ dará error porque el logaritmo sólo está definido para $x > 0$.

```
plot(log(x), -2, 2)
```


```
verbose 0 (4069: plot.py, generate_plot_points) WARNING: When
plotting, failed to evaluate function at 100 points.
verbose 0 (4069: plot.py, generate_plot_points) Last error message:
''
```



Elegir un intervalo que no muestra la información relevante de la función. Es como si, al fotografiar la cara de una persona encuadráramos sólo su nariz, o al revés, encuadráramos desde un avión la ciudad en la que se encuentra.


5) Dibuja la función $f(x) = x^4 - 10x^3 + 2$. Primero la dibujamos en el intervalo $[-5, 5]$.

```
plot(x^4-10*x^3+2, -5, 5)
```



La función parece decreciente. Sin embargo, si la pintamos en $[-8, 12]$


```
plot(x^4-10*x^3+2, -8, 12)
```




6) Dibujar la gráfica de la función $h(x) = \frac{x+1}{x^2+x+2}$ en los intervalos $[-2, 2]$ y $[-50, 50]$.

SAGE variará la escala del eje Y para que la gráfica quepa en el dibujo. Hay que ser conscientes de este cambio de escala, pero normalmente no conviene modificarlo.

```
h(x)=(x+1)/(x^2+x+2); plot(h(x), -2, 2)
```



```
plot(h(x), -50, 50)
```



Algunas veces sí será necesario que las escalas de los ejes coincidan. En esos casos añadiremos la opción `aspect_ratio`. Es decir, la sintaxis quedará

```
plot(f(x), xmin, xmax, aspect_ratio=1)
```


Si queremos otras escalas, cambiamos el valor de `aspect_ratio`.

7) Dibujemos una circunferencia centrada en el origen y de radio 2.

Despejando en $x^2 + y^2 = 2^2$ obtenemos que la parte superior de la circunferencia corresponderá a la gráfica de la función $y = \sqrt{4 - x^2}$, y la inferior a $y = -\sqrt{4 - x^2}$ (con x entre -2 y 2, por supuesto).


Pintamos ambas gráficas:

```
plot((sqrt(4-x^2), -sqrt(4-x^2)), -2, 2)
```



Nota que, con el cambio de escala, la circunferencia parece una elipse. Lo solucionaremos con la opción `aspect_ratio=1`.

```
plot((sqrt(4-x^2), -sqrt(4-x^2)), -2, 2, aspect_ratio=1)
```



En el siguiente ejemplo, que ya hemos visto anteriormente, también podemos observar el efecto del cambio de escala

```
plot(sin(x), -2*pi, 2*pi, rgbcolor=(1,0,0), aspect_ratio=1) +  
plot(cos(x), -2*pi, 2*pi, rgbcolor=(0,1,0), aspect_ratio=1)
```




show(g1,g2,...) muestra los gráficos g_1, g_2, \dots combinados en uno sólo.

Ya vimos que la orden **plot** permitía dibujar varios gráficos simultáneamente. ¿Por qué usar **show**? Hay varias ocasiones donde usar **show** nos resultará más cómodo o incluso imprescindible. Aquí sólo nombramos una: cuando queramos pintar gráficos que están definidos en dominios distintos. Aunque usualmente también puede hacerse con **plot**, resulta más sencillo usar **show**. El caso más típico son las funciones definidas a trozos:

8) Dibujar $f(x) = \begin{cases} 3/2 + \sqrt{1-x} & x < 1 \\ 1 - \sqrt{x-1} & x \geq 1 \end{cases}$ en el intervalo $(-2, 4)$. ¿Parece f continua?

Primero dibujamos la parte de la izquierda, luego la derecha y combinamos las dos (y se ve que f tiene una discontinuidad de salto en $x=1$):

```
gr1=plot(3/2+sqrt(1-x),-2,1); gr2=plot(1-sqrt(x-1),1,4)
show(gr1+gr2)
```



3.3. Límites de Funciones

La sintaxis de las órdenes que resuelven límites de funciones se presenta a continuación:

`limit(función, x=a)` calcula el límite de la función de variable **x**, indicada por su expresión analítica, cuando la variable **x** tiende al valor **a**.

`limit(función, x=a, dir="minus")` calcula el límite de la función de variable **x**, indicada por su expresión analítica, cuando la variable **x** tiende al valor **a** por la izquierda.

`limit(función, x=a, dir="plus")` calcula el límite de la función de variable **x**, indicada por su expresión analítica, cuando la variable **x** tiende al valor **a** por la derecha.

Para que una función tenga límite en un punto, es necesario que el límite por la izquierda y el límite por la derecha en ese punto coincidan. De no coincidir, el límite de la función en ese punto no existe.

Podemos calcular también límites en $+\infty$ y $-\infty$.

A continuación se presentan varios ejercicios relativos al cálculo de límites de funciones, algunos de los cuales se ilustrarán con las correspondientes gráficas. El uso de los gráficos es aconsejable siempre que surjan dudas acerca de los resultados.


1) Calcular $\lim_{x \rightarrow 1} \frac{1}{1-x} - \frac{3}{1-x^3}$ y representar gráficamente la función y el límite.

```
f(x)=1/(1-x)-3/(1-x^3); limit(f(x),x=1)
```

-1

Ahora representamos la función, añadiendo la recta horizontal **y=-1**, que nos ayuda a ver mejor el límite. Pedimos que la función se dibuje en color rojo y la recta **y=-1** en verde.

```
plot(f(x),0,2,rgbcolor=(1,0,0))+plot(-1,0,2,rgbcolor=(0,1,0))
```




2) Calcular $\lim_{x \rightarrow \infty} \log(x) - \log(2x-1)$ y representar gráficamente la función y el límite.

```
reset("f"); f(x)=log(x)-log(2*x-1); limit(f(x),x=oo)
```

-log(2)

```
plot((f(x),-log(2)),1,100)
```



3) Calcular el límite de la función $f(x) = e^{1/x}$ cuando $x \rightarrow 0$.

```
reset("f"); f(x)=e^(1/x); limit(f(x),x=0)
```


und

Se da cuenta de que no existe. Calculemos los límites laterales.

```
limit(f(x),x=0,dir="minus"); limit(f(x),x=0,dir="plus")
```

0
+Infinity

```
plot(f(x), -25, 25, ymin=0, ymax=2)
```



Luego la función no tiene límite cuando $x \rightarrow 0$.

4) Calcular el límite de la función $f(x) = \sin(1/x)$ cuando $x \rightarrow 0$.

```
reset("f"); f(x)=sin(1/x); limit(f(x),x=0)
```

```
ind
```


```
limit(f(x),x=0,dir="minus"); limit(f(x),x=0,dir="plus")
```

```
ind
```

```
ind
```

Vemos que en $x=0$ no existen los límites por la izquierda ni por la derecha. Veamos una gráfica de la función en torno al punto $x=0$.

```
plot(f(x), -1, 1)
```



3.4. Derivación

• Concepto de derivada

Una función $f(x)$ definida en un entorno del punto $x=a$, se dice derivable en a si existe el límite

$$\lim_{h \rightarrow 0} \frac{f(a+h) - f(a)}{h} = f'(a).$$

El valor del límite, si existe, se designa por $f'(a)$ y se llama derivada de f en el punto a . Si f es derivable en todos los puntos de su dominio, se dice simplemente que es derivable.

La recta tangente a la gráfica de f en el punto $(a, f(a))$ es $y - f(a) = f'(a)(x - a)$

1) Estudiar la derivabilidad de la función $f(x) = x \sin(1/x)$ si $x \neq 0$ y $f(x) = 0$ si $x = 0$.

Estudiamos en primer lugar la derivabilidad en el punto $x=0$.

```
reset("f"); f(x)=x*sin(1/x); h=var('h'); f(h)
```

```
h*sin(1/h)
```

```
limit((f(h)-f(0))/h,h=0)
```

```
Traceback (click to the left of this block for traceback)
```

```
...
```

```
RuntimeError: power::eval(): division by zero
```


El límite no existe, ya que la función $\sin(1/x)$ oscila infinitas veces entre -1 y 1 cuando $x \rightarrow 0$. Luego concluimos que $f'(0)$ no existe.

Hallemos la derivada en un punto $x=a$ distinto de cero y representemos la función:

```
reset("a"); a=var('a'); limit((f(a+h)-f(a))/h,h=0)
```

```
-cos(1/a)/a + sin(1/a)
```

```
plot(f(x), -1/10, 1/10)
```



• Cálculo de derivadas

SAGE ofrece varias funciones que permiten el cálculo de derivadas:

`diff(f(x))` derivada de la función **f(x)** respecto a su variable **x**.

`diff(f(x),x,4)` derivada cuarta.

`diff(g(x,y),y)` derivada de **g(x,y)** con respecto a la variable **y**.

`derivative` funciona exactamente igual.

1) Calcular la derivada de la función **f(x)=log(sen(2x))**.

```
reset("f"); f(x)=log(sin(2*x)); diff(f(x))
2*cos(2*x)/sin(2*x)
```

2) Calcular la derivada de la función $f(x) = \sqrt{\frac{-1+x^2}{2+x^2}}$ y simplificarla si se puede.

```
reset("f"); f(x)=sqrt((x^2-1)/(x^2+2)); derivative(f(x))
-((x^2 - 1)*x/(x^2 + 2)^2 - x/(x^2 + 2))/sqrt((x^2 - 1)/(x^2 + 2))
simplify(_)
-sqrt(x^2 + 2)*((x^2 - 1)*x/(x^2 + 2)^2 - x/(x^2 + 2))/sqrt(x^2 -
derivative(f(x)).simplify_radical()
3*sqrt(x^2 + 2)*x/(sqrt(x - 1)*sqrt(x + 1)*(x^4 + 4*x^2 + 4))
derivative(f(x)).simplify_rational()
3*sqrt(x^2 + 2)*x/(sqrt(x^2 - 1)*(x^4 + 4*x^2 + 4))
derivative(f(x)).simplify_full()
3*sqrt(x^2 + 2)*x/(sqrt(x - 1)*sqrt(x + 1)*(x^4 + 4*x^2 + 4))
```


• Interpretación geométrica de la derivada

Ejemplo: Dibujar la gráfica de **f(x)=x(x²-1)**, su tangente en **x=0** y las cuerdas que pasan por **x=0** y por otro punto **d≠0**.

```
reset("f"); d=var('d'); a=0
```

```
f(x)=x*(x^2-1); g(x)=diff(f(x)); tangen(x)=f(0)+g(0)*(x-0)
s(d)=f(a)+(f(d)-f(a))/(d-a)*(x-a)
```

```
gr1=plot(tangen(x),-1,1,rgbcolor=(1,0,0),thickness=2)
gr2=plot(f(x),-1,1,rgbcolor=(0,1,0),thickness=2)
show(gr1+gr2+plot([s(-1),s(-0.8),s(-0.6),s(-0.4),s(-0.2)]))
```



Vemos como según **d** se acerca a 0, la recta secante tiende a convertirse en la recta tangente.